Задатак

Електроенергетски трансформатор је у колу познатих параметара, $R_1 = R$, $R_2 = 2\,R$, $L_1 = L$, $L_2 = 4\,L$, $k = \frac{1}{2}$, $u_g(t) = U\,\vartheta(t)$, $i_1(t_0^-) = I_{01}$, $t_0 = 0$.

(6) Одредити струју секундара i_2 и њен домен.

(6) Одредити напон отворене везе u и његов домен.

(3) Нацртати график напона u.

Обележити осе графика, координатни почетак, пресеке и додире графика са осама, и тачке екстремума.



Решење

Задатак се може решити на више начина. Прво ћемо решити задато електрично коло у временском домену. Поставићемо систем једначина кола, нећемо уводити посебне променљиве за сваки приступ, написаћемо сажети систем једначина кола. На пример, нећемо уводити посебну ознаку за напон напонског извора већ ћемо сматрати да је тај напон u_g. Такође, сматраћемо да је напон отпорника R_1 једнак $R_1\,i_{R1}$ а да је напон отпорника R_2 једнак $R_2\,i_2$. За редне везе приступа ћемо користити једну променљиву за струју, а за паралелне везе једну променљиву за напон.

Систем једначина кола је

$$i_{R1} = i_1 + i_2$$

$$u_g = R_1\,i_{R1} + u_1$$

$$u = R_2\,i_2 + u_2$$

$$u_1 = L_1\,\frac{di_1}{dt} + L_{12}\,\frac{di_2}{dt}$$

$$u_2 = L_{12}\,\frac{di_1}{dt} + L_2\,\frac{di_2}{dt}$$

$$u = u_1$$

чини га шест једначина, а променљиве су i_1, i_2, i_{R1}, u, u_1, u_2. Заменом задатих вредности елемената, $R_1 = R$, $R_2 = 2\,R$, $L_1 = L$, $L_2 = 4\,L$, $k = \frac{1}{2}$, добија се $L_{12} = k\,\sqrt{L_1\,L_2} = L$ и систем једначина кола се своди на

$$i_{R1} = i_1 + i_2$$

$$u_g = R\,i_{R1} + u_1$$

$$u = 2\,R\,i_2 + u_2$$

$$u_1 = L\,\frac{di_1}{dt} + L\,\frac{di_2}{dt}$$

$$u_2 = L\,\frac{di_1}{dt} + 4\,L\,\frac{di_2}{dt}$$

$$u = u_1$$

Систему једначина кола се придружују природни почетни услови, напони кондензатора и струје калемова у почетном тренутку t-нула-минус, $i_1(t_0^-) = I_{01}$, $i_2(t_0^-) = I_{02} = 0$. Почетни тренутак је $t_0 = 0$.

Систем једначина кола решавамо за $t > t_0$ са почетним условима у t-нула-плус. За одређивање потребних почетних услова у t_0^+ су подесне једначине стања. Променљиве стања су струје i_1, i_2, и ако одредимо њих све остале променљиве ће се једноставно одредити из алгебарских једначина. На пример, за познате i_1, i_2 из прве једначине се добија i_{R1}, па из друге u_1, па из последње u, па из треће u_2.

Стање електричног кола је информација у посматраном тренутку времена која, заједно са познатим побудама, омогућава да се одреди понашање кола после посматраног тренутка времена.

Струје калемова и напони кондензатора чине стање кола; то су ***променљиве стања***. Напони кондензатора и струје калемова не могу да се тренутно промене ако у колу нема делта-импулса напона и струја. Каже се да ове величине памте (меморишу) стање кола.

Једначине стања су једначине по променљивама стања и побудама написане у договореном облику, у Кошијевој нормалној форми. Са десне стране једначине стања је линеарна нехомогена комбинација променљивих стања, а нехомогени део зависи од побуда и параметара. Са леве стране једначине стања је први извод по времену променљиве стања или, ако постоје алгебарске везе променљивих стања и побуда, нула. Број једначина стања је једнак броју динамичких елемената.

У овом задатку постоје два динамичка елемента, два калема, тако да је општи облик једначина стања

$$Z_1 \frac{di_1}{dt} = A_{11}\, i_1 + A_{12}\, i_2 + B_1\, u_{\mathrm{g}}$$

$$Z_2 \frac{di_2}{dt} = A_{21}\, i_1 + A_{22}\, i_2 + B_2\, u_{\mathrm{g}}$$

$Z_1, Z_2 \in \{0, 1\}$ узимају у обзир алгебарске дегенерације једначина стања.

Коефицијенти $A_{11}, A_{12}, A_{21}, A_{22}, B_1, B_2$ су функције параметара електричног кола R_1, R_2, L_1, L_2, k.

Систем једначина стања се може одредити на разне начине, али је за ручно решавање "помоћу папира и оловке" најједноставније увести смене првих извода променљивих стања, $dx_1 / dt = D_{\mathrm{x1}}$, $dx_2 / dt = D_{\mathrm{x2}}$, …, и решити тако добијени систем алгебарских једначина по променљивама $D_{\mathrm{x1}}, D_{\mathrm{x2}}$, … . Променљиве $x_1, x_2, \ldots$, су струје калемова и напони кондензатора. Када се одреде променљиве стања, струје калемова и напони кондензатора, сви остали напони и струје се одређују из њих решавањем **алгебарских** једначина.

Уводимо смене $\frac{di_1}{dt} = D_1$, $\frac{di_2}{dt} = D_2$, и решавамо тако добијен систем алгебарских једначина по променљивама D_1, D_2. Конкретно, систем алгебарских једначина

$$i_{\mathrm{R1}} = i_1 + i_2$$

$$u_{\mathrm{g}} = R\, i_{\mathrm{R1}} + u_1$$

$$u = 2\, R\, i_2 + u_2$$

$$u_1 = L\, D_1 + L\, D_2$$

$$u_2 = L\, D_1 + 4\, L\, D_2$$

$$u = u_1$$

има решење

$$D_1 = \frac{-R}{L}\, i_1 + \frac{-R}{3\,L}\, i_2 + \frac{1}{L}\, u_{\mathrm{g}}$$

$$D_2 = \frac{-2\,R}{3\,L}\, i_2$$

тако да је систем једначина стања

$$\frac{di_1}{dt} = \frac{-R}{L}\, i_1 + \frac{-R}{3\,L}\, i_2 + \frac{1}{L}\, u_{\mathrm{g}}$$

$$\frac{di_2}{dt} = \frac{-2R}{3L}\, i_2$$

или, у матричном облику,

$$\frac{d}{dt}\begin{bmatrix} i_1 \\ i_2 \end{bmatrix} = \begin{bmatrix} \frac{-R}{L} & \frac{-R}{3L} \\ 0 & \frac{-2R}{3L} \end{bmatrix}\begin{bmatrix} i_1 \\ i_2 \end{bmatrix} + \begin{bmatrix} \frac{1}{L}u_g \\ 0 \end{bmatrix}$$

Једначине стања у општем случају чине нехомоген систем линеарних диференцијалних једначина првог реда са константним коефицијентима.

Ред кола је број диференцијалних једначина у систему једначина стања. Ред кола је једнак или мањи од броја динамичких елемената.

Ред електричног кола у овом случају је два. Можемо задати два независна почетна услова за промењиве стања. Кажемо да коло има два степена слободе у погледу задавања почетних услова.

Користићемо ***став о почетним условима***, који прихватамо без доказа:
Посматрајмо временски непромењиво линеарно електрично коло без независних извора.
Ако је ред кола једнак броју динамичких елемената, онда су струје калемова и напони кондензатора непрекидни у почетном тренутку кола.

У задатку је број динамичких елемената једнак реду колу па су промењиве стања непрекидне функције времена у почетном тренутку, $i_1(t_0^+) = i_1(t_0) = i_1(t_0^-)$, $i_2(t_0^+) = i_2(t_0) = i_2(t_0^-)$.

Почетне услове за изводе промењивих стања добијамо из система једначина стања стављањем да је $t = t_0^+$.

$$\frac{di_1}{dt}(t_0^+) = \frac{-R}{L}\, i_1(t_0^+) + \frac{-R}{3L}\, i_2(t_0^+) + \frac{1}{L}\, u_g(t_0^+)\ ,\quad \frac{di_1}{dt}(t_0^+) = \frac{-R}{L}\, i_1(t_0^-) + \frac{-R}{3L}\, i_2(t_0^-) + \frac{1}{L}\, u_g(t_0^+)$$

$$\frac{di_2}{dt}(t_0^+) = \frac{-2R}{3L}\, i_2(t_0^+)\ ,\quad \frac{di_2}{dt}(t_0^+) = \frac{-2R}{3L}\, i_2(t_0^-)$$

У овом случају је друга једначина стања хомогена, само по једној промењивој, па је можемо решити са датим природним почетним условом $i_2(0^-) = 0$.

$$\frac{di_2}{dt} + \frac{2R}{3L}\, i_2 = 0\ ,\ \ i_2(t) = i_2(0^+)\, e^{\frac{-2R}{3L}t},$$

$$i_2(0^+) = i_2(0^-) = 0 \quad \Rightarrow \quad i_2(t) = 0,\ \ t \geq 0,\quad \text{непознато за}\ \ t < 0$$

Који је домен, област дефинисаности, струје секундара као функције времена? Предисторија електричног кола није позната, јер није познато како су настали почетни услови, тако да је $i_2(t)$ непознато за $t < t_0$. Домен је $t_0 \leq t < +\infty$ и на њему је $i_2(t)$ непрекидна функција времена.

Претходни резултат, $i_2 = 0$, можемо користити да се упрости прва једначина стања

$$\frac{di_1}{dt} = \frac{-R}{L}\, i_1 + \frac{1}{L}\, u_g$$

и да се она реши са природним почетним условом $i_1(0^-) = I_{01}$.

$$\frac{di_1}{dt} + \frac{R}{L}\, i_1 = \frac{1}{L}\, U\, \vartheta(t)\ ,$$

$$i_1(0^+) = i_1(0^-) = I_{01} \quad \Rightarrow \quad i_1(t) = I_{01}\, e^{\frac{-R}{L}t} + \frac{U}{R}\left(1 - e^{\frac{-R}{L}t}\right)\vartheta(t)\ ,$$

$$t \geq 0\ ,\quad \text{непознато за}\ \ t < 0$$

Струја примара је збир струје услед почетне енергије (почетних услова) и струје услед побуде (екситације).

Одзив на побуду је помножен Хевисајдовом функцијом, а одзив на почетне услове није. Потпуни одзив је непрекидна функција времена.

Хевисајдова функција $\vartheta(t)$, јединична одскочна функција, је дефинисана у сваком тренутку времена $-\infty < t < +\infty$. Каузална је, $\vartheta(t) = 0$, $t < 0$, Прекидна је у нули, $\vartheta(0^-) = 0$, $\vartheta(0) = 0$, $\vartheta(0^+) = 1$. Једнака је јединици за позитивно време, $\vartheta(t) = 1$, $t > 0$.

Који је домен, област дефинисаности, струје примара као функције времена? Предисторија електричног кола није позната, јер није познато како су настали почетни услови, тако да је $i_1(t)$ непознато за $t < t_0$. Домен је $t_0 \le t < +\infty$ и на њему је $i_1(t)$ непрекидна функција времена.

Тражени напон u се може одредити из алгебарских једначина

$$i_{\mathrm{R1}} = i_1 + i_2$$

$$u_{\mathrm{g}} = R\, i_{\mathrm{R1}} + u_1$$

$$u = 2\, R\, i_2 + u_2$$

$$u = u_1$$

из којих се добија

$$u = u_{\mathrm{g}} - R(i_1 + i_2)\,, \quad u(t) = -R\, I_{01}\, e^{\frac{-R}{L}\, t} + U\, e^{\frac{-R}{L}\, t}\, \vartheta(t)\,, \quad t \ge 0\,, \quad \text{непознато за } t < 0$$

Који је домен, област дефинисаности, напона u као функције времена?
Предисторија електричног кола није позната, јер није познато како су настали почетни услови, тако да је $u(t)$ непознато за $t < t_0$.
Домен је $t_0 \le t < +\infty$ и на њему је $u(t)$ дефинисана функција времена, али је непрекидна на отвореном интервалу $t_0 < t < +\infty$, и прекидна у почетном тренутку t_0.

Која је вредност овог напона у почетном тренутку?
Вредност је $u(0) = -R\, I_{01}$.
Подсетимо се да смо Хевисајдову одскочну функцију дефинисали тако да је у нули једнака нули, $\vartheta(0) = 0$.

Прерачунавање почетних услова је додатан напор у решавању електричног кола. Тежња да се избегне решавање диференцијалних једначина, и прерачунавање почетних услова, условила је тражење нових, делотворнијих и учинковитијих, поступака одређивања одзива, као што је метод анализе електричних кола Лапласовом трансформацијом.

Аутоматизована симболичка рачунарска анализа

```
In[1]:= $Assumptions = {L > 0, R > 0};

In[2]:= slog =
          {Di1 → D[Subscript[i, 1][t], t], Di2 → D[Subscript[i, 2][t], t], i1 → Subscript[i, 1],
           i2 → Subscript[i, 2], I01 → Subscript[Style["I", Italic], "01"],
           I02 → Subscript[Style["I", Italic], "02"], L1 → Subscript[L, 1],
           L2 → Subscript[L, 2], L12 → Subscript[L, 12],
           R1 → Subscript[R, 1], R2 → Subscript[R, 2],
           Ue → Subscript[Style["U", Italic], "e"], ug → Subscript[u, "g"]};

In[3]:= okvir[x_] := x // Style[#, 24] & // Framed[#, FrameStyle → Cyan] & // TraditionalForm

In[4]:= okvirY[x_] := x // Style[#, 24] & // Framed[#, FrameStyle → Yellow] & // TraditionalForm

In[5]:= zamenaLIT = {k → 1/2, L1 → L, L2 → 4 L};
```

In[6]:=
```
zamena =
  zamenaLIT ~ Join ~ {L12 → k √(L1 L2) /. zamenaLIT, R1 → R, R2 → 2 R, I02 → 0} // Simplify
```

Out[6]= $\left\{k \to \frac{1}{2}, L1 \to L, L2 \to 4 L, L12 \to L, R1 \to R, R2 \to 2 R, I02 \to 0\right\}$

Једначине стања

In[7]:=
```
jednacineD = {iR1 == i1 + i2, ug == R1 iR1 + u1,
   u == R2 i2 + u2, u1 == L1 Di1 + L12 Di2, u2 == L12 Di1 + L2 Di2, u == u1}
```

Out[7]=
```
{iR1 == i1 + i2, ug == iR1 R1 + u1, u == i2 R2 + u2,
 u1 == Di1 L1 + Di2 L12, u2 == Di1 L12 + Di2 L2, u == u1}
```

In[8]:=
```
jednacinei1i2 = Eliminate[jednacineD, {iR1, u, u1, u2}]
```

Out[8]=
```
Di1 L1 == Di1 L12 - Di2 L12 + Di2 L2 + i2 R2 && i1 R1 == -Di1 L12 - Di2 L2 - i2 R1 - i2 R2 + ug
```

In[9]:=
```
odzivDi1Di2 = Solve[jednacinei1i2, {Di1, Di2}] // FullSimplify
```

Out[9]= $\left\{\left\{Di1 \to \frac{i2\, L2\, R1 + i1\, (-L12 + L2)\, R1 - i2\, L12\, (R1 + R2) + (L12 - L2)\, ug}{L12^2 - L1\, L2},\; Di2 \to \frac{i1\, (L1 - L12)\, R1 - i2\, L12\, R1 + i2\, L1\, (R1 + R2) + (-L1 + L12)\, ug}{L12^2 - L1\, L2}\right\}\right\}$

In[10]:=
```
odzivD = Collect[First[odzivDi1Di2 /. Rule → Equal], {i1, i2}]
```

Out[10]= $\left\{Di1 == \frac{i1\, (-L12 + L2)\, R1}{L12^2 - L1\, L2} + \frac{i2\, (L2\, R1 - L12\, (R1 + R2))}{L12^2 - L1\, L2} + \frac{(L12 - L2)\, ug}{L12^2 - L1\, L2},\; Di2 == \frac{i1\, (L1 - L12)\, R1}{L12^2 - L1\, L2} + \frac{i2\, (-L12\, R1 + L1\, (R1 + R2))}{L12^2 - L1\, L2} + \frac{(-L1 + L12)\, ug}{L12^2 - L1\, L2}\right\}$

In[11]:=
```
Column[odzivD] /. slog // okvirY
```

Out[11]//TraditionalForm=

$$\begin{aligned} i_1'(t) &= \frac{(L_{12}-L_2)\, u_g}{L_{12}^2-L_1\, L_2} + \frac{i_1\, (L_2-L_{12})\, R_1}{L_{12}^2-L_1\, L_2} + \frac{i_2\, (L_2\, R_1-L_{12}\, (R_1+R_2))}{L_{12}^2-L_1\, L_2} \\ i_2'(t) &= \frac{(L_{12}-L_1)\, u_g}{L_{12}^2-L_1\, L_2} + \frac{i_1\, (L_1-L_{12})\, R_1}{L_{12}^2-L_1\, L_2} + \frac{i_2\, (L_1\, (R_1+R_2)-L_{12}\, R_1)}{L_{12}^2-L_1\, L_2} \end{aligned}$$

In[12]:=
```
Column[odzivD /. zamena] /. slog // okvir
```

Out[12]//TraditionalForm=

$$\begin{aligned} i_1'(t) &= \frac{u_g}{L} - \frac{i_1\, R}{L} - \frac{i_2\, R}{3\, L} \\ i_2'(t) &= -\frac{2\, i_2\, R}{3\, L} \end{aligned}$$

Анализа помоћу унилатералне Лапласове трансформације

Обележимо побуду са $u_g = U_e\, \vartheta(t)$ да би симбол U сачували за аутоматску симболичку анализу помоћу рачунара.

In[13]:=
```
ug = Ue HeavisideTheta[t];
```

In[14]:=
```
jednacine = {IR1 == I1 + I2, Ug == R1 IR1 + U1, U == R2 I2 + U2,
   U1 == L1 (s I1 - I01) + L12 (s I2 - I02), U2 == L12 (s I1 - I01) + L2 (s I2 - I02), U == U1};
```

In[15]:=
```
sZamena = {I1 → Subscript[Style["I", {Underlined, Italic}], 1][Style[s, Underlined]],
   I2 → Subscript[Style["I", {Underlined, Italic}], 2][Style[s, Underlined]], I01 →
    Subscript[Style["I", Italic], "01"], I02 → Subscript[Style["I", Italic], "02"],
   IR1 → Subscript[Style["I", {Underlined, Italic}], R1][Style[s, Underlined]],
   L1 → Subscript[Style["L", Italic], 1], L2 → Subscript[Style["L", Italic], 2],
   L12 → Subscript[Style["L", Italic], 12], R1 → Subscript[Style["R", Italic], 1],
   R2 → Subscript[Style["R", Italic], 2], s → Style[s, Underlined],
   U1 → Subscript[Style["U", {Underlined, Italic}], 1][Style[s, Underlined]],
   U2 → Subscript[Style["U", {Underlined, Italic}], 2][Style[s, Underlined]],
   Ue → Subscript[Style["U", Italic], "e"],
   Ug → Subscript[Style["U", {Underlined, Italic}], "g"][Style[s, Underlined]],
   U → Style["U", {Underlined, Italic}][Style[s, Underlined]], Rule → Equal};
```

Комплексан систем једначина кола у области унилатералне Лапласове трансформације је

In[16]:=
```
Column[jednacine] /. sZamena // TraditionalForm
```

Out[16]//TraditionalForm=

$$\underline{I}_{\mathrm{R1}}(\underline{s}) = \underline{I}_1(\underline{s}) + \underline{I}_2(\underline{s})$$
$$\underline{U}_{\mathrm{g}}(\underline{s}) = R_1\, \underline{I}_{\mathrm{R1}}(\underline{s}) + \underline{U}_1(\underline{s})$$
$$\underline{U}(\underline{s}) = R_2\, \underline{I}_2(\underline{s}) + \underline{U}_2(\underline{s})$$
$$\underline{U}_1(\underline{s}) = L_1\, (\underline{s}\, \underline{I}_1(\underline{s}) - I_{01}) + L_{12}\, (\underline{s}\, \underline{I}_2(\underline{s}) - I_{02})$$
$$\underline{U}_2(\underline{s}) = L_{12}\, (\underline{s}\, \underline{I}_1(\underline{s}) - I_{01}) + L_2\, (\underline{s}\, \underline{I}_2(\underline{s}) - I_{02})$$
$$\underline{U}(\underline{s}) = \underline{U}_1(\underline{s})$$

In[17]:=
```
odzivOpsti = Solve[jednacine, {I1, I2, IR1, U, U1, U2}];
```

Комплексан одзив, у општем случају, се добија решавањем комплексног система једначина кола, који је систем алгебарских линеарних једначина.

In[18]:=
```
Column[odzivOpsti // First // FullSimplify] /. sZamena // TraditionalForm
```

Out[18]//TraditionalForm=

$$\underline{I}_1(\underline{s}) = \frac{-L_{12}\, I_{01}\, (L_{12}\, \underline{s}+R_1)+L_1\, I_{01}\, (L_2\, \underline{s}+R_1+R_2)+L_2\, R_1\, (-I_{02})+L_{12}\, (R_1+R_2)\, I_{02}+(L_2-L_{12})\, \underline{s}\, \underline{U}_{\mathrm{g}}(\underline{s})+R_2\, \underline{U}_{\mathrm{g}}(\underline{s})}{R_1\, ((L_1+L_2-2\, L_{12})\, \underline{s}+R_2)+\underline{s}\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}$$

$$\underline{I}_2(\underline{s}) = \frac{(L_{12}-L_1)\, R_1\, I_{01}+I_{02}\, (-L_{12}\, R_1+L_2\, R_1+L_{12}^2\, (-\underline{s})+L_1\, L_2\, \underline{s})+(L_1-L_{12})\, \underline{s}\, \underline{U}_{\mathrm{g}}(\underline{s})}{R_1\, ((L_1+L_2-2\, L_{12})\, \underline{s}+R_2)+\underline{s}\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}$$

$$\underline{I}_{\mathrm{R1}}(\underline{s}) = \frac{L_1\, I_{01}\, (L_2\, \underline{s}+R_2)+L_{12}^2\, (-\underline{s})\, I_{01}+I_{02}\, (L_{12}\, R_2+L_{12}^2\, (-\underline{s})+L_1\, L_2\, \underline{s})+(L_1+L_2-2\, L_{12})\, \underline{s}\, \underline{U}_{\mathrm{g}}(\underline{s})+R_2\, \underline{U}_{\mathrm{g}}(\underline{s})}{R_1\, ((L_1+L_2-2\, L_{12})\, \underline{s}+R_2)+\underline{s}\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}$$

$$\underline{U}(\underline{s}) = \frac{(L_{12}^2-L_1\, L_2)\, R_1\, \underline{s}\, (I_{01}+I_{02})-R_2\, R_1\, (L_1\, I_{01}+L_{12}\, I_{02})+\underline{s}\, \underline{U}_{\mathrm{g}}(\underline{s})\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}{R_1\, ((L_1+L_2-2\, L_{12})\, \underline{s}+R_2)+\underline{s}\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}$$

$$\underline{U}_1(\underline{s}) = \frac{(L_{12}^2-L_1\, L_2)\, R_1\, \underline{s}\, (I_{01}+I_{02})-R_2\, R_1\, (L_1\, I_{01}+L_{12}\, I_{02})+\underline{s}\, \underline{U}_{\mathrm{g}}(\underline{s})\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}{R_1\, ((L_1+L_2-2\, L_{12})\, \underline{s}+R_2)+\underline{s}\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}$$

$$\underline{U}_2(\underline{s}) = \frac{R_1\, I_{01}\, (-L_{12}\, R_2+L_{12}^2\, \underline{s}-L_1\, L_2\, \underline{s})+(L_{12}^2-L_1\, L_2)\, (R_1+R_2)\, \underline{s}\, I_{02}+L_2\, R_1\, R_2\, (-I_{02})+\underline{s}\, \underline{U}_{\mathrm{g}}(\underline{s})\, (L_{12}\, R_2+L_{12}^2\, (-\underline{s})+L_1\, L_2\, \underline{s})}{R_1\, ((L_1+L_2-2\, L_{12})\, \underline{s}+R_2)+\underline{s}\, (L_1\, (L_2\, \underline{s}+R_2)-L_{12}^2\, \underline{s})}$$

In[19]:=
```
odzivUg = First[odzivOpsti] /. zamena // Simplify;
```

Заменом вредности елемената добија се комплексан одзив у посебном случају.

In[20]:=
```
Column[odzivUg] /. sZamena // TraditionalForm
```

Out[20]//TraditionalForm=

$$\underline{I}_1(\underline{s}) = \frac{L\, I_{01} + \underline{U}_g(\underline{s})}{L\, \underline{s} + R}$$
$$\underline{I}_2(\underline{s}) = 0$$
$$\underline{I}_{R1}(\underline{s}) = \frac{L\, I_{01} + \underline{U}_g(\underline{s})}{L\, \underline{s} + R}$$
$$\underline{U}(\underline{s}) = \frac{L\, \underline{s}\, \underline{U}_g(\underline{s}) - L\, R\, I_{01}}{L\, \underline{s} + R}$$
$$\underline{U}_1(\underline{s}) = \frac{L\, \underline{s}\, \underline{U}_g(\underline{s}) - L\, R\, I_{01}}{L\, \underline{s} + R}$$
$$\underline{U}_2(\underline{s}) = \frac{L\, \underline{s}\, \underline{U}_g(\underline{s}) - L\, R\, I_{01}}{L\, \underline{s} + R}$$

Комплексна побуда је

In[21]:=
```
Ugs = LaplaceTransform[ug, t, s]; Ugs /. sZamena // TraditionalForm
```

Out[21]//TraditionalForm=

$$\frac{U_e}{\underline{s}}$$

In[22]:=
```
odziv = odzivUg /. Ug → Ugs // Factor;
```

Комплексан одзив за конкретну побуду је

In[23]:=
```
Column[odziv] /. sZamena // okvir
```

Out[23]//TraditionalForm=

$$\boxed{\begin{aligned}
\underline{I}_1(\underline{s}) &= \frac{L\, \underline{s}\, I_{01} + U_e}{\underline{s}\,(L\, \underline{s} + R)}\\
\underline{I}_2(\underline{s}) &= 0\\
\underline{I}_{R1}(\underline{s}) &= \frac{L\, \underline{s}\, I_{01} + U_e}{\underline{s}\,(L\, \underline{s} + R)}\\
\underline{U}(\underline{s}) &= -\frac{L\,(R\, I_{01} - U_e)}{L\, \underline{s} + R}\\
\underline{U}_1(\underline{s}) &= -\frac{L\,(R\, I_{01} - U_e)}{L\, \underline{s} + R}\\
\underline{U}_2(\underline{s}) &= -\frac{L\,(R\, I_{01} - U_e)}{L\, \underline{s} + R}
\end{aligned}}$$

In[24]:=
```
{I1s, I2s, U1s, U2s, Us} = {I1, I2, U1, U2, U} /. odziv
```

Out[24]=

$$\left\{\frac{I01\, L\, s + Ue}{s\,(R + L\, s)},\ 0,\ -\frac{L\,(I01\, R - Ue)}{R + L\, s},\ -\frac{L\,(I01\, R - Ue)}{R + L\, s},\ -\frac{L\,(I01\, R - Ue)}{R + L\, s}\right\}$$

Одзив (временски одзив, потпун одзив), тренутне вредности, за $t > t_0$, односно за $t > 0$, је

In[25]:=
```
i1t = InverseLaplaceTransform[I1s, s, t] // Simplify // Collect[#, ⅇ-] &
```

Out[25]=

$$\frac{e^{-\frac{R\,t}{L}}\,(I01\, R - Ue)}{R} + \frac{Ue}{R}$$

In[26]:=
```
i2t = InverseLaplaceTransform[I2s, s, t]
```

Out[26]= 0

In[27]:=
```
u1t = InverseLaplaceTransform[U1s, s, t] // Simplify // Collect[#, ⅇ-] &
```

Out[27]=

$$e^{-\frac{R\,t}{L}}\,(-I01\, R + Ue)$$

In[28]:=
```
u2t = InverseLaplaceTransform[U2s, s, t] // Simplify // Collect[#, ⅇ-] &
```

Out[28]= $e^{-\frac{R\,t}{L}}\,(-I01\,R + Ue)$

In[29]:=
```
ut = InverseLaplaceTransform[Us, s, t] // Simplify // Collect[#, ⅇ-] &
```

Out[29]= $e^{-\frac{R\,t}{L}}\,(-I01\,R + Ue)$

In[30]:=
```
Row[{Subscript[i, 1][t] == i1t, ",   ", t > 0} /. slog] // okvirY
```

Out[30]//TraditionalForm=

$$\boxed{i_1(t) = \frac{e^{-\frac{R\,t}{L}}\,(R\,I_{01} - U_{\mathrm{e}})}{R} + \frac{U_{\mathrm{e}}}{R},\;\; t > 0}$$

In[31]:=
```
Row[{Subscript[i, 2][t] == i2t, ",   ", t ≥ 0} /. slog] // okvir
```

Out[31]//TraditionalForm=

$$\boxed{i_2(t) = 0,\;\; t \geq 0}$$

In[32]:=
```
Row[{Subscript[u, 1][t] == u1t, ",   ", t > 0} /. slog] // okvirY
```

Out[32]//TraditionalForm=

$$\boxed{u_1(t) = e^{-\frac{R\,t}{L}}\,(U_{\mathrm{e}} - R\,I_{01}),\;\; t > 0}$$

In[33]:=
```
Row[{Subscript[u, 2][t] == u2t, ",   ", t > 0} /. slog] // okvirY
```

Out[33]//TraditionalForm=

$$\boxed{u_2(t) = e^{-\frac{R\,t}{L}}\,(U_{\mathrm{e}} - R\,I_{01}),\;\; t > 0}$$

In[34]:=
```
Row[{u[t] == ut, ",   ", t > 0} /. slog] // okvir
```

Out[34]//TraditionalForm=

$$\boxed{u(t) = e^{-\frac{R\,t}{L}}\,(U_{\mathrm{e}} - R\,I_{01}),\;\; t > 0}$$

Одзив на конкретну побуду је дефинисан за $t \geq t_0$. Додаћемо множилац који садржи Хевисајдову функцију сабирцима који чине одзив на побуду. Препознаћемо сабирке у којима се појављује U_{e}.

In[35]:=
```
Row[{u[t] == (ut /. Ue → 0) + ϑ[t] (ut /. I01 → 0), ",   ", t ≥ 0} /. slog] // okvir
```

Out[35]//TraditionalForm=

$$\boxed{u(t) = U_{\mathrm{e}}\,\vartheta(t)\,e^{-\frac{R\,t}{L}} - R\,I_{01}\,e^{-\frac{R\,t}{L}},\;\; t \geq 0}$$

Нацртаћемо одзив само за $t \geq t_0$. Није познато како су настали почетни услови, тако да није познат одзив за негативно време.

У посебном случају, када је $U_{\mathrm{e}} = R\,I_{01}$, добија се $u(t) = 0$. Нацртаћемо график за општи случај $U_{\mathrm{e}} \neq R\,I_{01}$.

In[36]:=
```
Plot[ut/(Ue - R I01) /. (R t)/L → x, {x, 0, 5},
  AxesLabel → {(R t)/L /. slog // Style[#, FontFamily → "TimesNewRoman"] &,
    u[t]/(U - R I01) /. slog // Style[#, FontFamily → "TimesNewRoman"] &},
  PlotStyle → Thick, Epilog → {Text["(0,0)", {0.25, 0.05}]}]
```

Out[36]=


Потпун (комплетан) одзив можемо раздвојити на два сабирка:
(1) одзив на почетне услове и
(2) одзив на побуду који садржи као множилац Хевисајдову јединичну одскочну функцију.

Домен потпуног одзива је $t \geq t_0$.

Предисторија електричног кола није позната јер не знамо како су настали почетни услови.
Не познајемо потпун одзив за $t < t_0$.

Mathematica

In[37]:= `$Version`

Out[37]= `11.2.0 for Microsoft Windows (64-bit) (September 11, 2017)`

In[38]:= `DateString[]`

Out[38]= `Sun 10 Nov 2019 14:46:59`

```
(%i1)   load("C:\\SALECx\\SALECx.mac") $
```
Dejan Tosic, SALECx 2019 v1.0
Symbolic Analysis of Linear Electric Circuits with Maxima

```
(%i2)   SALECxPrint: true $
```




```
(%i3)   assume(L>0,R>0);
```
(%o3) $[L>0, R>0]$

```
(%i4)   LIT_shema: [
         ["V", "Ug", 3, 0, Ug],
         ["R", "R1", 3, 1, R1],
         ["R", "R2", 1, 2, R2],
         ["K", "LIT", [1,0], [2,0], [L1,L2,L12], [Io1,Io2]]
        ],
        [R1=R, R2=2*R,L1=L,L2=4*L,L12=(1/2)*sqrt(L*4*L),Io2=0]$
```

```
(%i5)   LIT_response: SALECx(LIT_shema);
```
Symbolic Analysis of Linear Electric Circuits with Maxima
SALECx version 1.0, Prof. Dr. Dejan Tošić, tosic@etf.rs
Number of nodes excluding 0 node: 3
Electric circuit specification: $[[V, Ug, 3, 0, Ug], [R, R1, 3, 1, R], [R, R2, 1, 2, 2R], [K, LIT, [1,0], [2,0], [L, 4L, L], [Io1, 0]]]$
Supported element: [true, true, true, true]
Element values: $[Ug, R, 2R, [L, 4L, L]]$
Initial conditions: $[false, false, false, [Io1, 0]]$
MNA equations: $\left[\frac{V_1-V_3}{R}+\frac{V_1-V_2}{2R}+I_{LIT,1}=0, \frac{V_2-V_1}{2R}+I_{LIT,2}=0, \frac{V_3-V_1}{R}+I_{Ug}=0, V_2=4I_{LIT,2}Ls+I_{LIT,1}Ls-Io1\,L, V_1=I_{LIT,2}Ls+I_{LIT,1}Ls-Io1\,L, V_3=Ug\right]$
MNA variables: $[V_1, V_2, V_3, I_{LIT,2}, I_{LIT,1}, I_{Ug}]$

(LIT_response) $\left[V_1=-\frac{Io1\,L\,R-L\,Ug\,s}{L\,s+R}, V_2=-\frac{Io1\,L\,R-L\,Ug\,s}{L\,s+R}, V_3=Ug, I_{LIT,2}=0, I_{LIT,1}=\frac{Ug+Io1\,L}{L\,s+R}, I_{Ug}=-\frac{Ug+Io1\,L}{L\,s+R}\right]$

```
(%i6)   transpose(LIT_response);
```
(%o6)
$$\begin{bmatrix} V_1=-\frac{Io1\,L\,R-L\,Ug\,s}{L\,s+R} \\ V_2=-\frac{Io1\,L\,R-L\,Ug\,s}{L\,s+R} \\ V_3=Ug \\ I_{LIT,2}=0 \\ I_{LIT,1}=\frac{Ug+Io1\,L}{L\,s+R} \\ I_{Ug}=-\frac{Ug+Io1\,L}{L\,s+R} \end{bmatrix}$$

```
(%i7) Ugs: laplace(U*unit_step(t),t,s);
WARNING: DEFUN/DEFMACRO: redefining function SIMP-UNIT-STEP in
         C:\maxima-5.38.1\share\maxima\5.38.1_5_gdf93b7b_dirty\share\orthopo
         was defined in top-level
WARNING: DEFUN/DEFMACRO: redefining function SIMP-POCHHAMMER in
         C:\maxima-5.38.1\share\maxima\5.38.1_5_gdf93b7b_dirty\share\orthopo
         was defined in top-level
```

(Ugs) $\frac{U}{s}$

```
(%i8) LIT_response_Ug: LIT_response, Ug=Ugs, ratsimp;
```

(LIT_response_Ug) $[V_1 = \frac{L\,U - Io1\,L\,R}{L\,s+R}, V_2 = \frac{L\,U - Io1\,L\,R}{L\,s+R}, V_3 = \frac{U}{s}, I_{LIT,2} = 0, I_{LIT,1} = \frac{Io1\,L\,s+U}{L\,s^2+R\,s}, I_{Ug} = -\frac{Io1\,L\,s+U}{L\,s^2+R\,s}]$

```
(%i9) transpose(LIT_response_Ug);
```

(%o9) $$\begin{bmatrix} V_1 = \frac{L\,U - Io1\,L\,R}{L\,s+R} \\ V_2 = \frac{L\,U - Io1\,L\,R}{L\,s+R} \\ V_3 = \frac{U}{s} \\ I_{LIT,2} = 0 \\ I_{LIT,1} = \frac{Io1\,L\,s+U}{L\,s^2+R\,s} \\ I_{Ug} = -\frac{Io1\,L\,s+U}{L\,s^2+R\,s} \end{bmatrix}$$




```
(%i10) V1s: V[1], LIT_response_Ug;
```

(V1s) $\frac{L\,U - Io1\,L\,R}{L\,s+R}$

```
(%i11) u1t: ilt(V1s,s,t), expand;
```

(u1t) $U\,\%e^{-\frac{R\,t}{L}} - Io1\,R\,\%e^{-\frac{R\,t}{L}}$

```
(%i12) I1s: I["LIT",1], LIT_response_Ug;
```

(I1s) $\frac{Io1\,L\,s+U}{L\,s^2+R\,s}$

```
(%i13) i1t: ilt(I1s,s,t), expand;
```

(i1t) $-\frac{U\,\%e^{-\frac{R\,t}{L}}}{R} + Io1\,\%e^{-\frac{R\,t}{L}} + \frac{U}{R}$

(%i14) `collectterms(i1t,Io2,U);`

(%o14) $U\left(\frac{1}{R}-\frac{\%e^{-\frac{R\,t}{L}}}{R}\right)+Io1\,\%e^{-\frac{R\,t}{L}}$

(%i15) `I2s: I["LIT",2], LIT_response_Ug;`

(I2s) 0

(%i16) `i2t: ilt(I2s,s,t);`

(i2t) 0

(%i17) `Us: V1s;`

(Us) $\frac{L\,U-Io1\,L\,R}{L\,s+R}$

(%i18) `ut: u1t;`

(ut) $U\,\%e^{-\frac{R\,t}{L}}-Io1\,R\,\%e^{-\frac{R\,t}{L}}$